
Abstract. We explore two general mechanisms for producing pairing bi-
jections (bijective functions defined from N2 → N). The first mechanism,
using n-adic valuations results in parameterized algorithms generating a
countable family of distinct pairing bijections. The second mechanism,
using characteristic functions of subsets of N provides 2N distinct pair-
ing bijections. Mechanisms to combine such pairing functions and their
application to generate families of permutations of N are also described.
The paper uses a small subset of the functional language Haskell to
provide executable specifications of various the functions defined in a lit-
erate programming style. The self-contained Haskell code extracted from
the paper is available at http://logic.cse.unt.edu/tarau/research/

2013/infpair.hs.

1 Introduction

Definition 1 A pairing bijection is a bijection f : N× N→ N. Its inverse f−1

is called an unpairing bijection.

We are emphasizing here the fact that these functions are bijections as the
name pairing function is sometime used in the literature to indicate injective-only
functions from N× N to N.

Pairing bijections have been used in the first half of 19-th century by Cauchy
as a mechanism to express double summations as simple summations in series.
They have been made famous by their uses in the second half of the 19-th century
by Cantor’s work on foundations of set theory. Their most well known application
is to show that infinite sets like N and N × N have the same cardinality. A
classic use in the theory of recursive functions is to reduce functions on multiple
arguments to single argument functions. Reasons on why they are an interesting
object of study in terms of practical applications range from multi-dimensional
dynamic arrays to proximity search using space filling curves are described in
[14, 6, 4, 7].



Like in the case of Cantor’s original function f(x, y) = 1
2 (x+y)(x+y+1)+y,

pairing bijections have been usually hand-crafted by putting to work geometric
or arithmetic intuitions.

This paper introduces two general mechanisms for building infinite families of
pairing functions, using n-adic valuations (section 2) and characteristic functions
of subsets of N (section 3), followed by a discussion of related work (section 5)
and our conclusions (section 6).

It is easy to prove that there is an uncountable family of distinct pairing
bijections and it is even easier to generate a countable family of pairing func-
tions simply by modifying its result of a fixed pairing function with a reversible
operation (e.g XOR with a natural number, seen as the index of the family).
Therefore, we will point out the main reasons for why our more complex pairing
bijections are interesting.

The n-adic valuation based pairing functions will provide a general mecha-
nism for designing strongly asymmetric pairing functions, where changes in one
of the arguments have an exponential impact on the result.

The characteristic-function mechanism, while intuitively obvious, opens the
doors, in combination with a framework providing bijections between them and
arbitrary data-types [18], to custom-build arbitrarily intricate pairing functions
associated, for instance, to “interesting” sequences of natural numbers [15] or
binary expansions of real numbers.

We will use a subset of the non-strict functional language Haskell (seen as
an equational notation for typed λ-calculus) to provide executable definitions
of mathematical functions on N, pairs in N× N, subsets of N, and sequences of
natural numbers. We mention, for the benefit of the reader unfamiliar with the
language, that a notation like f x y stands for f(x, y), [t] represents sequences
of type t and a type declaration like f :: s -> t -> u stands for a function
f : s × t → u (modulo Haskell’s “currying” operation, given the isomorphism
between the function spaces s× t → u and s→ t → u). Our Haskell functions
are always represented as sets of recursive equations guided by pattern matching,
conditional to constraints (simple arithmetic relations following | and before the
= symbol). Locally scoped helper functions are defined in Haskell after the where
keyword, using the same equational style. The composition of functions f and
g is denoted f . g. It is also customary in Haskell, when defining functions in
an equational style (using =) to write f = g instead of f x = g x (“point-free”
notation). The use of Haskell’s “call-by-need” evaluation allows us to work with
infinite sequences, like the [0..] infinite list notation, corresponding to the set
N itself.

2 Deriving Pairing Bijections from n-adic valuations

We first overview a mechanism for deriving pairing bijections from one-solution
Diophantine equations. Let us observe that

Proposition 1 ∀z ∈ N+ = N− {0} the Diophantine equation

2x(2y + 1) = z (1)



has exactly one solution x, y ∈ N.

This follows immediately from the unicity of the decomposition of a natural
number as a multiset of prime factors. Note that a slight modification of equation
1 results in the pairing bijection originally introduced in [8, 3], seen as a mapping
between the pair (x, y) and z.

2x(2y + 1)− 1 = z (2)

We will generalize this mechanism to obtain a family of bijections between
N×N and N+ (and the corresponding pairing bijections between N×N and N)
by choosing an arbitrary base b instead of 2.

Definition 2 Given a number n ∈ N, n > 1, the n-adic valuation of a natural
number m is the largest exponent k of n, such that nk divides m. It is denoted
νn(m).

Note that the solution x of the equation (1) is actually ν2(z).
Our generalization proceeds as follows. We implement, for and arbitrary b ∈

N, the functions nAdicCons b and nAdicDeCons b, forming a bijection between
N× N and N+.

nAdicCons :: N→(N,N)→N

nAdicCons b (x,y’) | b>1 = (b^x)∗y where

q = y’ ‘div‘ (b-1)

y = y’+q+1

nAdicDeCons :: N→N→(N,N)

nAdicDeCons b z | b>1 && z>0 = (x,y’) where

hd n = if n ‘mod‘ b > 0 then 0 else 1+hd (n ‘div‘ b)

x = hd z

y = z ‘div‘ (b^x)

q = y ‘div‘ b

y’ = y-q-1

Using nAdicDeCons we define the head and tail projection functions nAdicHead
and nAdicTail:

nAdicHead, nAdicTail :: N→N→N

nAdicHead b = fst . nAdicDeCons b

nAdicTail b = snd . nAdicDeCons b

Intuitively, their correctnes follows from the fact that the quotient y, after
dividing z with the largest possible power bx is not divisible by b. We then
“rebase” it to base b − 1, and make z correspond to a unique pair of natural
numbers.

More precisely, non-divisibility of y by b, stated as y = bq + m, b > m > 0
can be rewritten as y − q − 1 = bq − q + m − 1, b > m > 0, or equivalently
y − q − 1 = (b − 1)q + (m − 1), b > m > 0 from where it follows that by
setting y′ = y − q − 1 and m′ = m − 1 we map z to a pair (y′,m′) such that



y′ = (b − 1)q + m′ and b − 1 > m′ ≥ 0. So we can transform (y,m) such that
y = bq + m with b > m > 0, into a pair (y′,m′), such that y′ = q(b − 1) + m′

with b − 1 > m′ ≥ 0. Note that the transformation works also in the opposite
direction with y′ = y − q − 1 giving y = y′ + q + 1, and with m′ = m− 1 giving
m = m′ + 1.

The following examples illustrate the operations for base 3:

*InfPair> nAdicCons 3 (10,20)

1830519

*InfPair> nAdicHead 3 1830519

10

*InfPair> nAdicTail 3 1830519

20

Note that nAdicHead n x computes the n-adic valuation of x, νn(x) while the
tail corresponds to the “information content” extracted from the quotient, after
division by νn(x).

Definition 3 We call the natural number computed by nAdicHead n x the n-
adic head of x ∈ N+, by nAdicTail n x the n-adic tail of x ∈ N+ and the
natural number in N+ computed by nAdicCons n (x,y) the n-adic cons of
x, y ∈ N.

By generalizing the mechanism shown for the equations 1 and 2, we derive
from nAdicDeCons and nAdicCons the corresponding pairing and unpairing bi-
jections nAdicPair and nAdicUnPair:

nAdicUnPair :: N→N→(N,N)

nAdicUnPair b n = nAdicDeCons b (n+1)

nAdicPair :: N→(N,N)→N

nAdicPair b xy = (nAdicCons b xy)-1

One can see that we obtain a countable family of bijections fb : N × N → N
indexed by b ∈ N, b > 1.

The following examples illustrate the work of these bijections for b = 3.

*InfPair> map (nAdicUnPair 3) [0..7]

[(0,0),(0,1),(1,0),(0,2),(0,3),(1,1),(0,4),(0,5)]

*InfPair> map (nAdicPair 3) it

[0,1,2,3,4,5,6,7]

Note the use of Haskell’s higher-order function “map”, that applies the function
nAdicUnPair 3 to a list of elements and collects the results to a list, and the
special value “it”, standing for the previously computed result.

2.1 Deriving bijections between N and [N]

For each base b>1, we can also obtain a pair of bijections between natural
numbers and lists of natural numbers in terms of nAdicHead, nAdicTail and
nAdicCons:



nat2nats :: N→N→[N]

nat2nats _ 0 = []

nat2nats b n | n>0 =
nAdicHead b n : nat2nats b (nAdicTail b n)

nats2nat :: N→[N]→N

nats2nat _ [] = 0

nats2nat b (x:xs) = nAdicCons b (x,nats2nat b xs)

The following example illustrate how they work:

*InfPair> nat2nats 3 2012

[0,2,2,0,0,0,0]

*InfPair> nats2nat 3 it

2012

Using the framework introduced in [16, 18] and summarized in the Ap-
pendix, we can “reify” these bijections as Encoders between natural num-
bers and sequences of natural numbers (parameterized by the first argument
of nAdicHead and nAdicTail). Such Encoders can now be “morphed”, by using
the bijections provided by the framework, into various data types sharing the
same “information content” (e.g. lists, sets, multisets).

nAdicNat :: N→Encoder N

nAdicNat k = Iso (nat2nats k) (nats2nat k)

In particular, for k = 2, we obtain the Encoder corresponding to the Dio-
phantine equation (1)

nat :: Encoder N

nat = nAdicNat 2

The following examples illustrate these operations, lifted through the framework
defining bijections between datatypes, given in Appendix.

*InfPair> as (nAdicNat 3) list [2,0,1,2]

873

*InfPair> as (nAdicNat 7) list [2,0,1,2]

27146

*InfPair> as nat list [2,0,1,2]

300

*InfPair> as list nat it

[2,0,1,2]

2.2 Deriving new families of Encoders and Permutations of N

For each l, k ∈ N one can generate a family of permutations (bijections f : N→
N), parameterized by the pair (l,k), by composing nat2nats l and nats2nat

k.

nAdicBij :: N → N → N → N

nAdicBij k l = (nats2nat l) . (nat2nats k)



The following example illustrates their work on the initial segment [0..31] of
N:

*InfPair> map (nAdicBij 2 3) [0..31]

[0,1,3,2,9,5,6,4,27,14,15,8,18,10,12,7,81,41,42,

22,45,23,24,13,54,28,30,16,36,19,21,11]

*InfPair> map (nAdicBij 3 2) [0..31]

[0,1,3,2,7,5,6,15,11,4,13,31,14,23,9,10,27,63,

12,29,47,30,19,21,22,55,127,8,25,59,26,95]

It is easy to see that the following holds:

Proposition 2
(nAdicBij k l) ◦ (nAdicBij l k) ≡ id (3)

We can derive Encoders representing functions between N and sequences
of natural numbers, parameterized by a (possibly infinite) list of nAdicHead /

nAdicTail bases, by repeatedly applying the n-adic head, tail and cons operation
parameterized by the (assumed infinite) sequence ks:

nAdicNats :: [N]→Encoder N

nAdicNats ks = Iso (nat2nAdicNats ks) (nAdicNats2nat ks)

nat2nAdicNats :: [N]→N→[N]

nat2nAdicNats _ 0 = []

nat2nAdicNats (k:ks) n | n>0 =
nAdicHead k n : nat2nAdicNats ks (nAdicTail k n)

nAdicNats2nat :: [N]→[N]→N

nAdicNats2nat _ [] = 0

nAdicNats2nat (k:ks) (x:xs) =
nAdicCons k (x,nAdicNats2nat ks xs)

For instance, the Encoder nat’ corresponds to ks defined as the infinite sequence
starting at 2.

nat’ :: Encoder N

nat’ = nAdicNats [2..]

The following examples illustrate the mechanism:

*InfPair> as nat’ list [2,0,1,2]

1644

*InfPair> as list nat’ it

[2,0,1,2]

*InfPair> map (as nat’ nat) [0..15]

[0,1,2,3,4,7,6,5,8,19,14,15,12,13,10,9]

*InfPair> map (as nat’ nat) [0..15]

[0,1,2,3,4,7,6,5,8,19,14,15,12,13,10,9]

Note that functions like as nat’ nat illustrate another general mechanism for
defining permutations of N.



3 Pairing bijections derived from characteristic functions
of subsets of N

We start by connecting the bitstring representation of characteristic functions
to our bijective data transformation framework (overviewed in the Appendix).

3.1 The bijection between lists and characteristic functions of sets

The function list2bins converts a sequence of natural numbers into a charac-
teristic function of a subset of N represented as a string of binary digits. The
algorithm interprets each element of the list as the number of 0 digits before
the next 1 digit. Note that infinite sequences are handled as well, resulting in
infinite bitstrings.

list2bins :: [N]→[N]

list2bins [] = [0]

list2bins ns = f ns where

f [] = []

f (x:xs) = (repl x 0) ++ (1:f xs) where

repl n a | n ≤ 0 = []

repl n a = a:repl (pred n) a

The function bin2list converts a characteristic function represented as bit-
strings back to a list of natural numbers.

bins2list :: [N] → [N]

bins2list xs = f xs 0 where

f [] _ = []

f (0:xs) k = f xs (k+1)
f (1:xs) k = k : f xs 0

Together they provide the Encoder bins, that we will use to connect character-
istic functions to various data types.

bins :: Encoder [N]

bins = Iso bins2list list2bins

The following examples (where the Haskell library function take is used to re-
strict execution to an initial segment of an infinite list) illustrate their use:

*InfPair> list2bins [2,0,1,2]

[0,0,1,1,0,1,0,0,1]

*InfPair> bins2list it

[2,0,1,2]

*InfPair> take 20 (list2bins [0,2..])

[1,0,0,1,0,0,0,0,1,0,0,0,0,0,0,1,0,0,0,0]

*InfPair> bins2list it

[0,2,4,6]

The following holds:



Proposition 3 If M is a subset of N, the bijection as bins set returns the
bitstring associated to M and its inverse is the bijection as set bins.

Proof. Observe that the transformations are the composition of bijections be-
tween bitstrings and lists and bijections between lists and sets.

The following example illustrates this correspondence:

*InfPair> as bins set [0,2,4,5,7,8,9]

[1,0,1,0,1,1,0,1,1,1]

*InfPair> as set bins it

[0,2,4,5,7,8,9]

Note that, for convenient use on finite sets, the functions do not add the infinite
stream of 0 digits indicating its infinite stream of non-members, but we will add
it as needed when the semantics of the code requires it for representing accurately
operations on infinite sequences. We will use the same convention through the
paper.

3.2 Splitting and merging bitstrings with a characteristic function

Guided by the characteristic function of a subset of N, represented as a bitstring,
the function bsplit separates a (possibly infinite) sequence of numbers into two
lists: members and non-members.

bsplit :: [N] → [N] → ([N], [N])

bsplit _ [] = ([],[])

bsplit [] (n:ns) =
error ("bspilt provides no guidance at: "++(show n))

bsplit (0:bs) (n:ns) = (xs,n:ys) where

(xs,ys) = bsplit bs ns

bsplit (1:bs) (n:ns) = (n:xs,ys) where

(xs,ys) = bsplit bs ns

Guided by the characteristic function of a subset of N, represented as a
bitstring, the function bmerge merges two lists of natural numbers into one,
by interpreting each 1 in the characteristic function as a request to extract an
element of the first list and each 0 as a request to extract an element of the
second list.

bmerge :: [N] → ([N], [N]) → [N]

bmerge _ ([],[]) = []

bmerge bs ([],[y]) = [y]

bmerge bs ([x],[]) = [x]

bmerge bs ([],ys) = bmerge bs ([0],ys)

bmerge bs (xs,[]) = bmerge bs (xs,[0])

bmerge (0:bs) (xs,y:ys) = y : bmerge bs (xs,ys)

bmerge (1:bs) (x:xs,ys) = x : bmerge bs (xs,ys)

The following examples (trimmed to finite lists) illustrate their use:



*InfPair> bsplit [0,1,0,1,0,1] [10,20,30,40,50,60]

([20,40,60],[10,30,50])

*InfPair> bmerge [0,1,0,1,0,1] it

[10,20,30,40,50,60]

3.3 Defining pairing bijections, generically

We design a generic mechanism to derive pairing functions by combining the
data type transformation operation as with the bsplit and bmerge functions
that apply a characteristic function encoded as a list of bits.

genericUnpair :: Encoder t → t → N → (N, N)

genericUnpair xEncoder xs n = (l,r) where

bs = as bins xEncoder xs

ns = as bins nat n

(ls,rs) = bsplit bs ns

l = as nat bins ls

r = as nat bins rs

genericPair :: Encoder t → t → (N, N) → N

genericPair xEncoder xs (l,r) = n where

bs = as bins xEncoder xs

ls = as bins nat l

rs = as bins nat r

ns = bmerge bs (ls,rs)

n = as nat bins ns

Let us observe first that termination of this function depends on termination
of the calls to bsplit and bmerge, as illustrated by the following examples:

*InfPair> genericPair bins (cycle [0]) (10,20)

^CInterrupted.

*InfPair> genericUnpair bins (cycle [1]) 42

(^CInterrupted.

In this case, the characteristic functions given by cycle [0] or cycle [1] would
trigger an infinite search for a non-existing first 1 or 0 in bsplit and bmerge.

Clearly, this suggests restrictions on the acceptable characteristic functions.
We will now give sufficient conditions ensuring that the functions genericUnpair

and genericPair terminate for any values of their last arguments. Such restric-
tions, will enable them to define families of pairing functions parameterized by
characteristic functions derived from various data types.

Definition 4 We call block of digits occurring in a characteristic function any
(finite or infinite) contiguous sequence of digits.

Note that an infinite block made entirely of 0 (or 1) digits can only occur at the
end of the sequence defining the characteristic function, i.e. only if it exists a
number n such that the index of each member of the block is larger than n.



Proposition 4 If {an}n∈N is an infinite sequence of bits containing only finite
blocks of 0 and 1 digits, genericPair bins and genericUnPair bins define a
family of pairing bijections parameterized by {an}n∈N.

Proof. Having an alternation of finite blocks of 1s and 0s, ensures that, when
called from genericPair and genericUnPair, the functions bmerge and bsplit

terminate.

For instance, Morton codes [4] are derived by using a stream of alternating
1 and 0 digits (provided by the Haskell library function cycle)

bunpair2 = genericUnpair bins (cycle [1,0])

bpair2 = genericPair bins (cycle [1,0])

and working as follows:

*InfPair> map bunpair2 [0..10]

[(0,0),(1,0),(0,1),(1,1),(2,0),

(3,0),(2,1),(3,1),(0,2),(1,2),(0,3)]

*InfPair> map bpair2 it

[0,1,2,3,4,5,6,7,8,9,10]

Proposition 5 If {an}n∈N is an infinite sequence of non-decreasing natural
numbers only containing finite blocks of equal values, the functions genericPair
set and genericUnPair set define a family of pairing bijections parameterized
by {an}n∈N.

Proof. Observe that the sequence represents canonically an infinite set such that
its complement is also an infinite non-decreasing sequence. Therefore, the asso-
ciated characteristic function will have an alternation of finite blocks of 1 and 0

digits, inducing a pairing/unpairing bijection.

The bijection bpair k and its inverse bunpair k are derived from a set

representation (implicitly morphed into a characteristic function).

bpair k = genericPair set [0,k..]

bunpair k = genericUnpair set [0,k..]

Note that for k = 2 we obtain exactly the bijections bpair2 and bunpair2
derived previously, as illustrated by the following example:

*InfPair> map (bunpair 2) [0..10]

[(0,0),(1,0),(0,1),(1,1),(2,0),(3,0),

(2,1),(3,1),(0,2),(1,2),(0,3)]

*InfPair> map (bpair 2) it

[0,1,2,3,4,5,6,7,8,9,10]

We conclude with a similar result for lists:

Proposition 6 If {an}n∈N is an infinite sequence of natural numbers only con-
taining finite blocks of 0s, the functions genericPair list and genericUnPair

list define a family of pairing bijections parameterized by {an}n∈N.



Proof. It follows from Prop. 5 by observing that such sequences are transformed
into infinite sets represented as non-decreasing sequences only containing finite
blocks of equal values.

The Appendix discusses a few more examples of such pairing functions and
visualizes a few space-filling curves associated to them.

Proposition 7 There are 2N pairing functions defined using characteristic func-
tions of sets of N.

Proof. Observe that a characteristic function corresponding to a subset of N
containing an infinite block of 0 or 1 digits necessarily ends with the block.
Therefore, by erasing the block we can put such functions in a bijection with a
finite subset of N. Given that there are only a countable number of finite subsets
of N, the cardinality of the set of the remaining subsets’ characteristic functions
is 2N.

Proposition 8 The infinite sequence of binary digits representing any irrational
number corresponds to a unique pairing function.

Proof. Observe that no periodic sequence can be present in the binary represen-
tation of an irrational number, and in particular the representation cannot end
with an infinite block of 0s or 1s.

4 Unpairing Distance: Comparing Pairing Functions

We will now introduce a mechanism to compare pairing functions (and their
suitability for a particular application).

Unpairing maps a natural number into a pair that can be seen as a point
in the 2-dimensional plane. We can borrow the distance between such pairs and
define it as the unpairing distance between two numbers:

unpairDist unpairF distF n m = d where

p1 = unpairF n

p2 = unpairF m

d = distF p1 p2

An interesting derived property would be the length of the path from 0 to n,
seen as the sum of the distances between consecutive numbers up to n.

pathLen _ _ 0 = 0

pathLen unpairF distF n | n> 0 = d where

ps = map unpairF [0..n]

ds1 = tail ps

ds2 = init ps

ds = zipWith distF ds1 ds2

d = sum ds

An example of distance (which has the advantage of returning natural number
values is the Manhattan distance:



manhattanDist (a,b)(c,d) = abs (a-c) + abs(b-d)

We will use it to define the “Manhattan length” of a natural number n, for a
given unpairing function unpairF as:

manhattanLen unpairF n = pathLen unpairF manhattanDist n

We can specialize our unpairing functions to obtain n-adic and bunpair dis-
tances and length associated to a number.

nAdicDist p m n = unpairDist (nAdicUnPair p) manhattanDist n m

bunpairDist p m n = unpairDist (bunpair p) manhattanDist n m

nAdicLen p n = manhattanLen (nAdicUnPair p) n

bunpairLen p n = manhattanLen (bunpair p) n

The following examples illustrate these concepts:

*InfPair> nAdicDist 2 1000 2000

500

*InfPair> nAdicDist 3 1000 2000

224

*InfPair> bunpairDist 2 1000 2000

42

*InfPair> bunpairDist 3 1000 2000

62

*InfPair> map (nAdicLen 2) [0..15]

[0,1,3,6,10,12,15,21,28,31,35,41,48,52,57,68]

*InfPair> map (bunpairLen 2) [0..15]

[0,1,3,4,6,7,9,10,14,15,17,18,20,21,23,24]

As the examples show, distances and path lengths tend to be larger for asym-
metric unpairing functions like nAdicUnPair that favor the their second output
value, than for symmetric ones like bunpair 2.

An interesting open problem would be finding the unpairing function giv-
ing the shortest average path length for numbers up to n, for a given distance
function.

Note also that these concepts can be easily lifted to n-dimensional spaces, by
replacing pairing functions with untupling functions that split a natural number
bijectively into a tuple of n natural numbers.

5 Related Work

Pairing functions have been used in work on decision problems as early as [8, 3,
10, 11, 13, 12]. There are about 16400 Google documents referring to the original
“Cantor pairing function” among which we mention the surprising result that,
together with the successor function it defines a decidable subset of arithmetic
[2]. An extensive study of various pairing functions and their computational



properties is presented in [1, 14]. They are also related to 2D-space filling curves
(Z-order, Gray-code and Hilbert curves) [6, 4, 5, 7]. Such curves are obtained
by connecting pairs of coordinates corresponding to successive natural numbers
(obtained by applying unpairing operations). They have applications to spatial
and multi-dimensional database indexing [6, 4, 5, 7] and symbolic arbitrary length
arithmetic computations [19]. Note also that bpair 2 and bunpair 2 are the
same as the functions defined in [9] and also known as Morton-codes, with uses
in indexing of spatial databases [6].

6 Conclusion

We have described two mechanisms for generating countable and uncountable
families of pairing / unpairing bijections. The mechanism involving n-adic val-
uations is definitely novel, and we have high confidence (despite of their obvi-
ousness) that the characteristic function-based mechanisms are novel as well, at
least in terms of their connections to list, set or multiset representations provided
by the implicit use of our bijective data transformation framework [16].

Given the space constraints, we have not explored the natural extensions to
more general tupling / untupling bijections (defined between Nk and N) as well as
bijections between finite lists, sets and multisets that can be derived quite easily,
using the data transformation framework given in the Appendix. For the same
reasons we have not discussed specific applications of these families of pairing
functions, but we foresee interesting connections with possible cryptographic
uses (e.g “one time pads” generated through intricate combinations of members
of these families).

The ability to associate such pairing functions to arbitrary characteristic
functions as well as to their equivalent set, multiset, list representations provides
convenient tools for inventing and customizing pairing / unpairing bijections, as
well as the related tupling / untupling bijections and those defined between
natural numbers and sequences, sets and multisets of natural numbers.

We hope that our adoption of the non-strict functional language Haskell
(freely available from haskell.org), as a complement to conventional mathe-
matical notation, enables the empirically curious reader to instantly validate our
claims and encourage her/him to independently explore their premises and their
consequences.

Appendix

An Embedded Data Transformation Language

We will describe briefly the embedded data transformation language used in this
paper as a set of operations on a groupoid of isomorphisms. We refer to ([16,
17]) for details.

The Groupoid of Isomorphisms We implement an isomorphism between
two objects X and Y as a Haskell data type encapsulating a bijection f and its
inverse g.

X Y
................................................................................................................................................................................................................................................................................... ............

f = g−1

...............................................................................................................................................................................................................................................................................................

g = f−1

We will call the from function the first component (a section in category the-
ory parlance) and the to function the second component (a retraction) defining
the isomorphism. The isomorphisms are naturally organized as a groupoid.

data Iso a b = Iso (a→b) (b→a)

from (Iso f _) = f

to (Iso _ g) = g

compose :: Iso a b → Iso b c → Iso a c

compose (Iso f g) (Iso f’ g’) = Iso (f’ . f) (g . g’)

itself = Iso id id

invert (Iso f g) = Iso g f

Assuming that for any pair of type Iso a b, f ◦ g = idb and g ◦ f = ida, we can
now formulate laws about these isomorphisms.

The data type Iso has a groupoid structure, i.e. the compose operation, when
defined, is associative, itself acts as an identity element and invert computes the
inverse of an isomorphism.

The Hub: Sequences of Natural Numbers To avoid defining n(n−1)
2 iso-

morphisms between n objects, we choose a Hub object to/from which we will
actually implement isomorphisms.

Choosing a Hub object is somewhat arbitrary, but it makes sense to pick a
representation that is relatively easy convertible to various others and scalable
to accommodate large objects up to the runtime system’s actual memory limits.

We will choose as our Hub object sequences of natural numbers. We will
represent them as lists i.e. their Haskell type is [N].



type N = Integer

type Hub = [N]

We can now define an Encoder as an isomorphism connecting an object to Hub

type Encoder a = Iso a Hub

together with the combinator “as”, providing an embedded transformation lan-
guage for routing isomorphisms through two Encoders.

as :: Encoder a → Encoder b → b → a

as that this x = g x where Iso _ g = compose that (invert this)

The combinator “as” adds a convenient syntax such that converters between A
and B can be designed as:

a2b x = as B A x

b2a x = as A B x

Hub

A B

...........
...........
...........
...........
...........
...........
...........
...........
...........
...........
...........
...........
...........
...........
..............
............

b

..........
..........

..........
..........

..........
..........

..........
..........

..........
..........

..........
..........

..........
..........

..............................

a−1

....................................................................................................................................................................
....
............

b−1

.............................................................................................................................................................. ........
....

a

................................................................................................................................................................................................................................................................................... ............
a2b = as B A

...............................................................................................................................................................................................................................................................................................

b2a = as A B

Given that [N] has been chosen as the root, we will define our sequence data
type list simply as the identity isomorphism on sequences in [N].

list :: Encoder [N]

list = itself

The Encoder mset for multisets of natural numbers is defined as:

mset :: Encoder [N]

mset = Iso mset2list list2mset

mset2list, list2mset :: [N]→[N]

mset2list xs = zipWith (-) (xs) (0:xs)

list2mset ns = tail (scanl (+) 0 ns)

The Encoder set for sets of natural numbers is defined as:

set :: Encoder [N]

set = Iso set2list list2set

set2list, list2set :: [N]→[N]

list2set = (map pred) . list2mset . (map succ)

set2list = (map pred) . mset2list . (map succ)

Note that these converters between lists, multisets and sets make no assumption
about finiteness of their arguments and therefore they can used in a non-strict
language like Haskell on infinite objects as well.



Examples of pairing functions derived from characteristic functions

The function syracuse is used in an equivalent formulation of the Collatz conjec-
ture. Interestingly, it can be computed using the nAdicTail which results after
dividing a number n with µ2(n). Note that we derive our pairing function di-
rectly from the list representation of the range of this function as genericPair
and genericUnpair implicitly construct the associated characteristic function.

syracuse :: N→N

syracuse n = nAdicTail 2 (6∗n+4)

nsyr 0 = [0]

nsyr n = n : nsyr (syracuse n)

syrnats = map syracuse [0..]

syrpair = genericPair list syrnats

syrunpair = genericUnpair list syrnats

Figures 1 and 2 show the “Z-order” (Morton code) path connecting succes-
sive values in the range of the function bunpair 2 and bunpair 3. Figures 3
and 4 show the path connecting the values in the range of unpairing functions
associated, respectively to the Syracuse function and the binary digits of π. In-
terestingly, at a first glance, some regular patterns emerge even in the case of
such notoriously irregular characteristic functions.

Fig. 1. Path connecting values of bunpair 2



Fig. 2. Path connecting values of bunpair 3

Fig. 3. Path connecting values of an unpairing bijection based on the Syracuse function

Fig. 4. Path connecting values of an unpairing bijection based on binary digits of π


